NOAA Technical Memorandum ERL WPL-113

DEPARTMENT OF COMMERCE · UNITED STATES OF AMERICA

BETA: A PROGRAM FOR CALCULATING AND ARCHIVING BACKSCATTERING PROFILES TAKEN WITH THE NOAA COHERENT LIDAR SYSTEM

Wave Propagation Laboratory
Boulder, Colorado
November 1983

noaa NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION / Environmental Research Laboratories

NOAA Technical Memorandum ERL WPL-113

BETA: A PROGRAM FOR CALCULATING AND ARCHIVING BACKSCATTERING PROFILES TAKEN WITH THE NOAA COHERENT LIDAR SYSTEM

M. J. Post
J. F. Morrow
D. B. Jensen

Wave Propagation Laboratory
Boulder, Colorado
November 1983

UNITED STATES
DEPARTMENT OF COMMERCE

Malcolm Baldrige,
Secretary

NATIONAL OCEANIC AND
ATMOSPHERIC ADMINISTRATION

John V. Byrne,
Administrator

Environmental Research
Laboratories

Vernon E. Derr,
Director

NOTICE

Mention of a commercial company or product does not constitute an endorsement by NOAA Environmental Research Laboratories. Use for publicity or advertising purposes of information from this publication concerning proprietary products or the tests of such products is not authorized.

CONTENTS

	Page
ABSTRACT	1
INTRODUCTION	1
TECHNICAL BACKGROUND	2
RUNNING BETA	7
APPENDIX A	25
APPENDIX B	34
REFERENCES	44

BETA: A PROGRAM FOR CALCULATING AND ARCHIVING BACKSCATTERING PROFILES TAKEN WITH THE NOAA COHERENT LIDAR SYSTEM

M. J. Post, J. F. Morrow, and D. B. Jensen

ABSTRACT

We document the interactive program BETA to provide to users information about its purpose and its inputs and outputs. More emphasis is placed on running the program and understanding the raw and processed data than is placed on programing particulars.

INTRODUCTION

NOAA has developed a pulsed, coherent, infrared Doppler laser-radar (lidar) system (Post et al., 1982) for measuring atmospheric winds to ranges of 10-20 km. Pulses of 3-μs duration are transmitted at 10.0-Hz pulse repetition frequency (PRF), and a very small amount of energy from the outgoing pulse is scattered by aerosol particles back towards the lidar. By measuring the frequency shift (Doppler shift) of this backscattered signal, we can deduce the velocity component of the aerosols which is in line with (radial to) the laser beam. Since the aerosols are so small that they move "exactly" with the winds, we say we are measuring the radial wind component. By scanning the lidar beam in a conical pattern, it is possible to deduce the average speed and direction of the winds in the pattern.

The whole process depends on having enough aerosols to backscatter energy from the transmitted pulse, so there is a genuine need to document the backscattering properties of the aerosols and their statistical fluctuations. The backscattering coefficient β is a measure of the aerosol's backscattering ability. By definition it is the effective total cross-sectional area in m^2 of all the particles in one cubic meter of air, divided by 4π. The division by 4π assumes the energy is scattered isotropically into a 4π steradian solid angle (sphere), so β has units of $m^2/m^3/sr$ or $m^{-1}\ sr^{-1}$.

The most efficient way to measure β vs. altitude, or the β-profile, is to point the lidar beam vertically (angle from straight up = zenith angle = 0°), but any zenith angle is permissible.

We describe here the BETA program, which uses the recorded lidar signals and calculates the β-profile that had to exist in order to give rise to the signals. The program itself is listed in Appendix B.

TECHNICAL BACKGROUND

The lidar's data-taking system digitizes an analog voltage originating on the heterodyne signal detector and records this signal on 9-track digital magnetic tape. The digitizer has 8-bit resolution, a minimum sampling interval of 100 ns (maximum rate of 10 MHz) and a sampling capacity of 2048 consecutive points. The data are complex; that is, for each data point, two numbers are generated (R and Q) which are 90° out of phase (hence they are in "quadrature" and represent the real and imaginary parts of a complex number). Signal power is computed by squaring both R and Q and taking the sum of the squares.

The gain of the receiving chain is adjusted so that when no signal is present, the mean background noise level uses only 1-3 bits of the dynamic range of the digitizer, leaving 7-5 bits for the (additive) signal level. The maximum unsaturated signal plus noise is 128^2, or 16,384.

We know the relationship between the volume backscatter coefficient β and the system SNR. It is

$$SNR = \frac{\eta A_1 A_2 A_3 A_4 J a^2 \Pi c \tau \beta}{8 h\nu B R^2}$$

where η = Detector quantum efficiency

$A_1(R)$ = Gaseous absorption factor =

$$\exp -2 \int_o^R \sum \alpha_i(R')dR' \quad (i = \text{species})$$

$A_2(R)$ = Apodization factor

A_3 = Shot noise factor

A_4 = Round trip optical loss factor

a = Beam radius at primary (m)

c = Speed of light ($m\ s^{-1}$)

τ = Pulse duration (s)

β = Backscatter coefficient ($m^{-1}\ sr^{-1}$)

h = Planck's constant (J s)

ν = Laser frequency (s^{-1})

B = Bandwidth (s^{-1})

R = Range (m)

J = Energy per pulse (J)

The A factors are discussed in more detail by Lawrence (1983). The combination of η, A_3, and A_4 is variable ETA in the BETA program, and is set to a value of 0.48, assuming the following baseline calibration using a small spinning disc:

CW SNR = 43 dB

Range = 159 m

CW power = 0.5 w

Shot noise = 3.7 dB.

Many of the terms in the SNR equation vary, and must be set properly when BETA is run. Among these, A_1 accounts for round-trip atmospheric absorption of the CO_2 radiation by H_2O, CO_2, and other gases. One of five AFCRL absorption models (McClatchey et al., 1972) may be chosen (1 = tropical, 2 = midlatitude summer, 3 = midlatitude winter, 4 = subarctic summer, 5 = subarctic winter), but only models 2 or 3 are appropriate for U.S. data. We investigated the use of radiosonde measurements to better account for H_2O absorption effects, and found that the present models introduce no more than 2 dB error in β at extreme ranges (~20 km) and for extreme (unrealistic) mismatches between the models and the radiosonde-measured atmosphere. The mean errors between the midlatitude summer and winter models and the average radiosonde profiles for Denver increase monotonically from zero at 2 km altitude to 0.29 dB above 9 km altitude. A more complete description of continuum absorption by water vapor and its variability is given in Appendix A. A peculiarity we find is that the absorption coefficient is nearly constant with altitude above 10 km, despite decreasing density, owing to pressure and temperature effects on the CO_2 line shape and strength.

Another variable parameter is B, the system bandwidth. Noise power is proportional to B, but signal power is not a function of B if its frequency spectrum lies entirely within B. Typically, we set B to 1.5 MHz when the lidar is pointing vertically, since in this configuration we expect negligible radial velocity. When we take data at zenith angles other than zero, however, wind can cause Doppler shifts greater than 1.5 MHz, so we set the bandwidth to 10.0 MHz.

The last variable system parameter is J, the energy per pulse. We measure the mean output power with and without the TEA section turned on. The difference between powers is the average pulse power. Dividing this by the PRF (pulse repetition frequency) gives energy per pulse.

Sensitivity profiles for Boulder, Colo., in both summer and winter are shown in Fig. 1 for the indicated system parameters. The profiles were determined empirically by calculating β profile from system noise (telescope-blocked) files. We know from the "white" character of our system noise that false β values will appear 16% of the time for such files when we use a 1 km noise window and a 0.1-km altitude-averaging interval. Thus, the sensitivity curves represent the 84% confidence level for 0.1-km averaging intervals and the 92% level for 1.0-km intervals.

We use a "quality factor" to reject false signals by comparing the relative uncertainties in system noise power (N) and signal plus noise powers (S+N). The mean noise level ($\overline{N}$) in the noise window has a standard deviation σ_N; both are numerically determined in the program. For low signals, $(S+N) \approx \overline{N}$ and $\sigma_{S+N} \simeq \sigma_N = \sigma$. If the noise window contains ℓ_w data points, the altitude-averaging interval contains ℓ_β points, and the uncertainty σ in a mean power level is reduced by $\ell^{-1/2}$ when averaging over ℓ points, to detect signal $\overline{S}$ above noise $\overline{N}$ (where each is averaged over ℓ_β and ℓ_w points) we require $\overline{S} = (S+N) - \overline{N}$ to be greater than $\sigma(\ell_\beta{}^{-1/2} + \ell_w{}^{-1/2})$. We define the quality factor $Q = \overline{S}/\sigma$ and require $Q > (\ell_\beta{}^{-1/2} + \ell_w{}^{-1/2})$ for the observed data before they are accepted for inclusion in a β profile.



Figure 1.--System sensitivity curves for NOAA's coherent lidar, determined empirically using system noise and two different atmospheric absorption models. Range effect (R^2) is shown for reference only.

RUNNING BETA

To run the BETA program on the NOAA Eclipse, you must be in the LIDAR directory.

Type DIR DPØ:LIDAR.

If hard copies of the graphics output are desired, use the Tektronix terminal with attached hard copy unit, pre-warmed. Make sure the data tape is mounted and on the load point, and the tape drive is on line.

Type BETA to start the program.

Figures 2-10 show sequential outputs from the terminal screen and user responses to illustrate the discussion that follows.

When the screen clears and asks for the tape drive unit number,

Enter Ø or 1Ø to select unit the tape is mounted on.

The screen will then clear again and ask for calibration data.

Enter the most pertinent calibration data, which may be found on the calibration log sheets.

After you enter these data, the computer calculates and prints out the factor CAL, which is used to correct ETA. Beta values are multiplied by this factor near the end of the program to account for variations in calibration.

Enter the bandwidth in MHz from the tape log sheet next.

This is typically 10.0 for Doppler mode data or 1.5 for β mode data.

```
TAPE UNIT NUMBER 0
CW SNR (DB) ? 48.5
RANGE (M) ? 178
CW POWER (W) ? 1.2
SHOT  NOISE (DB) ? 2.5
CALIBRATION FACTOR IS        .707139

BANDWIDTH (IN MHZ) 1.5

ALTITUDE OF LIDAR (KM ASL) 1.604

INITIAL RANGE (KM) 1.891
END RANGE (KM) 30.450

WHICH ATMOSPHERIC ABSORPTION MODEL TO USE ?
1=TROPICAL
2=MIDLATITUDE SUMMER
3=MIDLATITUDE WINTER  3

FILE NUMBER ? 1
START RECORD NUMBER ? 1
NUMBER OF RECORDS TO PROCESS ? 200

ZENITH ANGLE OF SCANNER (DEGREES) ? 0

PER PULSE ENERGY (J) WAS ? .09

CONTINUE (1) OR GO BACK AND START OVER (0)? 1
```

Figure 2.--Sample terminal printouts and user responses for initialization of program.

Enter the lidar altitude above sea level (ASL) in kilometers. This permits properly initializing the absorption models with respect to sea level.

Enter beginning and ending ranges.

These values will set the vertical scaling of the plots to follow. Typically, the minimum useful beginning range is 1.5 km because the tail of the optical pulse saturates the detector. The maximum ending range is determined by the length of the record and the digitization rate. It is about 31.5 km for 2048 point seconds and a 10.0 MHz rate. Normally no signal is found in the stratosphere, or after 10 to 15 km range, but volcanic loading of the stratosphere has produced usable signals to 29 km range, and hard targets (mountains, towers, etc.) are visible at even greater ranges. You may choose a narrower range increment positioned anywhere as well, for example, to examine a cloud layer. For archiving β data, we want to average over 1-km altitude increments, centered on even kilometers above sea level. This requires setting the proper beginning range, which can be found with the aid of a sketch showing the altitude of the lidar ASL and its beam angle. The beginning altitude (and corresponding beginning range) will occur at a half integer value. Typical settings for the various field locations are written on a card near the terminal.

Enter the number of the atmospheric absorption model to be used.

Enter file and start record number to be processed. It is permissible to spin tape forward or backward.

Enter number of records to process in the file (this number must be less than or equal to the number of records in the file).

To compute altitudes ASL and to account for slant path absorption,

Enter the lidar beam angle from the zenith (vertical) in degrees, as indicated on the data sheet.

The computer will calculate the correct absorption profile and store it for use later, when β's are computed.

Enter the per-pulse energy from the data sheet.

Check for errors.

If no errors have been made in all your entries,

Enter 1 and the tape will be properly positioned.

If you made an error,

Enter Ø, and make all entries again.

Next, set the scaling for the graphic output by answering the appropriate questions on the terminal (see Fig. 3). Full (HI) scale of the digitizer is 0-20,000, but most interesting data are viewed on the 0-400 scale (LO). Also available is logarithmic scaling, 0-40 (HI) or 0-12 (LO) dB. Once set, scaling cannot be changed during a run.

Set switch Ø on the front panel of the computer. Turning the switch up selects plot averages, while turning it down selects single plots

Switch Ø may be changed at any time. Normally, start with it up, but occasionally flip it down during a run to see if the data are good, to find out what record in the file is currently being processed, or simply to avert

```
SET SWITCH (O): UP FOR AVERAGE PLOTS ONLY.
                DOWN FOR SINGLE RECORD PLOTS.

NOTE:  SWITCH (0) MAY BE CHANGED DURING PROCESSING.

THE FOLLOWING SCALING OPTIONS CANNOT BE CHANGED.
 LINEAR (1) OR DB (0) SCALING ? 1
 HI (1) OR LO (0) SCALING ? 0

PLOTTING FORMAT IS SET FOR:
    AVERAGE PLOTS ONLY
    X-AXIS 0-400 LINEAR

IF THIS FORMAT IS OK, ENTER 1. IF NOT, ENTER 0. 1
```

Figure 3.--Sample terminal printouts and user responses for scaling of graphical outputs.

boredom. It should be flipped back up immediately again, however, as processing slows to the graphical frame rate while it is down.

Enter 1 if you are satisfied with the scaling format.

Enter Ø if you wish to reselect the format.

Next, the file header is read by the computer and its contents are displayed (Fig. 4).

Check the variables to make sure you are working on the right file. If satisfied,

Strike any key and processing will commence.

The tape moves in spurts--once or twice per second. Nothing will appear on the screen unless switch Ø is depressed for single records (see Fig. 5) or until all records are processed. Then a time history of the total backscattered power (one point for each record's average power) is displayed (see Fig. 6). Laser power fluctuations, "losing lock," intrusion of clouds, etc., are easily seen on this plot. If you decide some bad records were included, deduce which ones (time in seconds and record number are in one-to-one correspondence). Note the bad records, if any, on the log sheet.

Make a hard copy of this plot to document data quality.

Strike the page key to clear the page.

To eliminate sections of bad records,

Enter Ø, which permits reprocessing the correct records.

```
FILE  1
DATE 11/ 2/83
TIME  9:34:26

  1     DIGITIZER RATE ( .08-10MHZ)               10.00
  2     DELAY TO 1ST SAMPLE (US)                    .15
  3     NUMBER OF SAMPLES PER PULSE             2048.00
  4     TRIGGER MODE                               3.00
  5     TRIGGER SPACING (MS)                     100.00
 10     NUMBER OF RECORDS DIGITIZED.            1000.00

PAUSE "STRIKE ANY KEY TO CONTINUE"
```

Figure 4.--Printout of sample tape header listing parameters of the current file.



Figure 5.--Sample single pulse profile. $R^2 + Q^2$ on horizontal axis, altitude (km) on vertical axis.



Figure 6.--Sample plot of signal power vs. time (or power vs. pulse number in file). Each point represents the average returned signal power for the pulse occurring at the indicated time. Display is useful for detecting signal variability throughout a file, whether due to atmospheric or system changes.

To continue, (when all records appear satisfactory),

Enter 1.

If you continue, the next plot will appear. It is a graph of uncorrected signal strength vs. altitude, averaged over all records (Fig. 7).

Make a hard copy of this plot for future reference and to help in choosing the "noise window" as discussed below.

Any anomalies on this plot, particularly in the magnitude and character of the noise, indicate problems. Judging the quality of the data in this manner is important, and can be learned only through experience.

Note bad files or bad portions of the profile on the log sheet to aid the next user.

Strike the page key to clear the screen, then

Enter 1 or Ø (print out or do not print out the raw data values (see Fig. 8)

To get a printout, the printer must be powered and on line.

Enter the beginning and ending altitudes for the "noise window," typically 1 to 2 km apart.

Choosing this window is one of the most critical inputs you make, for it affects the accuracy of the computed β profiles greatly. Choose this window after inspecting the raw data (previous plot). The window must include only noise and no signal. All the data points in the window are analyzed to find



Figure 7.--Sample plot of the average of 1000 single pulse profiles (as in Fig. 5). Incoherent averaging reduces noise fluctuations and permits signals to appear.

```
PRINT RAW DATA ?   YES = 1  NO = 0  0
BEG, END ALTITUDES (KM) FOR NOISE WINDOW ?  27,29

ALTITUDE RESOLUTION IN KM (1.005 FOR ARCHIVING) ?.3

Q-FACTOR MUST BE GREATER THAN  .35

PRINT OUT BETA PROFILE ?   YES = 1   NO = 0  0
```

Figure 8.--Sample terminal output and user responses for controlling processed data outputs, both tabular and graphical.

the mean noise level and the standard deviation of the noise. (A historical note: When the lidar system was first built, the mean noise versus time (or range) was not constant, and a stored, average noise profile had to be scaled and adjusted to match the noise level in the window. It was then subtracted from all the raw data to recover the signal power.) After the preamplifier was upgraded, the noise became nearly constant in time (range), permitting the mean noise level in the window to be subtracted from all data in the profile, a simpler operation. If you can find no window containing only noise, correct β's cannot be computed.

Enter the altitude increment for averaging, or the height resolution you desire on the β plot.

Choose 0.3 km for matching to the lidar pulse length (and hence to the inherent system resolution). However, when archiving you must average over 1.005 km increments. This makes the reported β values occur at even kilometers altitude above sea level (after roundoff errors), assuming you entered the correct beginning range.

From the mean noise in the window, its standard deviation, and the altitude increment over which you are averaging the signal, the computer calculates and prints out the "quality factor" discussed earlier. The signal must exceed this criterion in order for us to be more than 84% confident that we are indeed sensing signal power and not simply statistical fluctuations in the noise. If you desire the β profile to be printed out at this point,

Enter 1, provided the printer is on line and has power.

Enter Ø to proceed without printout.

Next the β profile is plotted (Fig. 9). An experienced user will see anomalies immediately, e.g., spikes of increasing magnitude with height, which indicate bad data or an improper choice of the noise window. The signal must exceed the quality factor in two consecutive altitude increments in order for β to be accepted for either. That is, no isolated β's will be accepted; this helps eliminate occasional noise spikes. If the β for an altitude increment is rejected either for this reason or for not passing the quality factor test, it is set to $1.0 \times 10^{-15}\ m^{-1}\ sr^{-1}$. If you are not happy with the β plot, you may try to improve it by choosing another noise window and/or altitude increment. To do this,

Strike the page key.

Choose new noise windows and/or altitude increments until you are satisfied that β's have been calculated correctly.

Figure 10 shows the same β profile as Fig. 9, but averaged over 1.0 km in preparation for archiving. When viewing the β-profiles, bear in mind the system sensitivity curves of Fig. 1. β's falling just below the curve or intermittant β points generally following it are suspect. Again, experience and comparison with the raw data profile are invaluable in judging the quality of the retrieved β profiles.

Once you are satisfied with the β profile,

Make a hard copy and store it with the other plots in the log sheet binder.

Strike the page key.



Figure 9.--Sample β profile output with 0.3-km height resolution.



Figure 10.--Sample β profile output (as in Fig. 9), with 1.0-km height resolution. $\overline{p}$

Decide whether to archive the data, provided the β values are positioned on even kilometers above sea level and are spaced 1 km apart.

Enter 1 to archive, Ø to continue without archiving.

It is highly desirable to keep the archivings sequential. When archived, the raw SNR's, quality factors, and β values are recorded as a single record on a disc file labeled "ARCH", together with many "housekeeping" factors such as date, tape file number, noise window, and absorption model. If you did archive,

Record the archiving immediately on the master archive list to ensure up-to-date documentation of this important file.

The screen will display the pertinent parameters for the master list. Another program, CPYARCH, copies the highest quality and statistically independent records from the ARCH file onto file SUPARCH for later processing. After recording,

Strike the page key.

Enter 1 to process other sections of the tape (essentially starting the program over again).

Enter Ø to stop the program.

By stopping, you initiate an automatic tape rewind and return the terminal to the RDOS command level. In this case,

Take the tape drive unit off line (press the "on line" button)

Off-load the tape by pressing the "rewind" button.

Dismount and store the tape.

Finally, return the log sheet binder to its proper shelf space.

APPENDIX A

EFFECTS OF VARIABLE ATMOSPHERE WATER VAPOR ON β-PROFILES

In Boulder, Colo., we use either the midlatitude summer (2) or midlatitude winter (3) model of AFCRL (McClatchey et al., 1972) to correct for the effects of atmospheric absorption when computing β-profiles. The major gaseous constituent of the atmosphere most likely to depart from these models on a given day is water vapor. This Appendix quantifies the statistical variability between our models and the real atmospheric water vapor absorption, (as computed from measurements of temperature, pressure, and dew point by radiosondes released from Denver, Colo.). While making these calculations, we noted that at $\lambda = 10.6$ μm, water vapor continuum absorption was more than 100 times greater than water vapor line absorption, so we ignored the latter in studying these effects.

If the water vapor profiles in the AFCRL model atmospheres are integrated, they amount to the various precipitable water vapor (W) amounts listed in Table AI. For Boulder, Colo., data taken at an elevation of 1.67 km, we eliminate the lower 1.67 km of water vapor from these models and are left with 1.30 cm for the midlatitude summer model and 0.42 for the midlatitude winter model.

TABLE AI--Precipitable water vapor in AFCRL models

(0 - 10 km altitude)

Model	W Precipitable Water (cm)
Tropical	4.15
Midlatitude summer	3.00
Midlatitude winter	0.86
Subarctic summer	2.10
Subarctic winter	0.42

TABLE AII--Precipitable water by month in Denver, Colorado

(1.6 - 7.5 km altitude)

Month	W Precip. Water (cm)	σ_W Std. Dev. (cm)
1	0.502	.220
2	0.476	.183
3	0.527	.195
4	0.696	.237
5	1.040	.344
6	1.444	.384
7	1.897	.485
8	1.813	.498
9	1.302	.422
10	0.886	.301
11	0.649	.223
12	0.499	.205

The profiles of water vapor in the models lead to the two-way absorption curves (continuum only) shown in Figs. A1 and A2.

A National Weather Service report prepared by Lott (1976) lists the monthly mean precipitable water vapor from the surface in Denver (1.611 km) to 400 mb (~7.5 km), along with its standard deviation. Essentially all water vapor is contained in this region. This listing is duplicated in Table AII. The summer average for June, July, and August is 1.72 cm, and the winter average for December, January and February is 0.49 cm; both are close to the model amounts.

From the radiosonde profiles of temperature, pressure, and dew point depression, we use Zuev (1972) to compute the water vapor partial pressure E at a given altitude Z:

$$E = E_o \, 10^{b\tau(c + \tau)}$$

where E_o = 6.11 mb

b = 7.5

c = 237.3°K

τ = dew point in °C.

Then the absorption coefficient, α_n (at level n) due to water vapor continuum can be computed, according to Roberts et al. (1976). The one-way transmission T_i to level i is then found by $T_i = e^{-x}$, where



Figures A1 and A2.--Two-way water vapor continuum absorption loss vs. altitude for the AFCRL summer (A1) and winter (A2) models for a vertically pointing lidar of 10.6 μm wavelength at an altitude of 1.67 km.

$$x = \sum_{n=1}^{i} \alpha_n \Delta Z_n \sec \theta$$

and

$$\theta = \text{zenith angle.}$$

Thus, the two-way absorption in dB is

$$10 \log T_i^2 = 8.686 \sum_{n=1}^{i} \alpha_n Z_n \sec \theta.$$

The FORTRAN code used for these calculations is listed at the back of this appendix, and is called H2OABS.

We selected 24 radiosonde profiles to analyze--two extremely dry ($W \sim 0.1$ cm), ten mean summer amounts ($W \sim 1.7$ cm), and ten mean winter amounts ($W \sim 0.5$ cm). For the ten summer profiles, the ratio of the standard deviation of W to mean W was 0.013; the same ratio was 0.04 for the winter profiles. Figures A1 and A2 show the mean two-way loss profiles and standard deviations, in contrast to the model losses. Note that maximum differences are small--0.3 dB for summer (7%) and 0.03 dB for winter (0.7%). Using Lott's values of standard deviations of W, we compute the standard deviations of the losses at the top of the loss profiles as ~0.1 dB for winter and ~0.4 dB for summer. These fluctuations reduce to zero along with the loss profile as altitude is decreased. Thus the AFCRL loss profiles are contained within the error bars of the radiosonde-derived loss profiles. Figures A3 and A4 show mean α profiles for the radiosonde sets.

Comparing the extreme cases with the models, we find maximum differences of -0.15 to 2.70 dB in the winter and -0.70 to 2.15 dB in the summer. But the larger winter difference (2.7 dB) is not likely to occur, since we are





Figures A3 and A4.--Calculated absorption coefficients vs. altitude for ten summer (A3) and ten winter (A4) sample radiosonde profiles, each profiles having nearly the Denver mean precipitable water vapor for that season. Error bars show standard deviations of the coefficient at each altitude.

contrasting a summer radiosonde-derived loss profile with a winter model, and such a high W case will not occur in the winter. Reasonable bounds to put on all differences are thus -1.0 to +2.0 dB for the Denver-Boulder area and zenith pointing.

Errors introduced at other locations by using AFGL models instead of radiosonde profiles are likely to be greater, since the Denver area is relatively dry. Errors at any location will increase dramatically with increasing zenith angles, but such errors can be derived for any location and geometry using the techniques described here.

Listing of H2OABS follows on next page.

```
C         SUBROUTINE H2OABS(ABSRP)
C   THIS SUBROUTINE FINDS THE H2O ABSORPTION FROM RAWIN DATA.
C   THE DATA IS TYPED INTO A FILE OF ARBITRARY NAME CONTAINED
C   IN ARRAY NFNAM.  THE FORMAT
C   OF THE DATA IS PRESSURE, ALTITUDE IN METERS, TEMPERATURE IN
C   DEGREES C, DEW POINT DEPRESSION.
C   THE PROGRAM THEN FINDS THE PARTIAL PRESSURE OF WATER VAPOR
C   (E) AND THE ABSORPTION DUE TO IT (SIGMA).
C   SINCE THE RAWIN DATA IS AT
C   RANDOM HEIGHTS AND WE WANT SPECIFIC INTERVALS, THE DO 20 LOOP
C   INTERPOLATES SIGMA LINEARLY.  WE CHECKED WHETHER IT WOULD BE
C   BETTER TO INTERPOLATE  BEFORE DOING CALCULATIONS
C   AND THE RESULTS AGREED TO WITHIN ONE PERCENT.
C
C         THE FORM OF THE EQUATIONS TO CALCULATE THE EXTINCTION
C         COEFFICIENT SIGMA IS FROM ROBERTS ET AL: I.R. CONTINUUM
C         ABSORPTION ETC., APPL. OPT. 15, P2085. EQUATIONS TO COMPUTE
C         WATER VAPOR DENSITY FROM RAWINSONDE OUTPUTS IS FROM ZUEV:
C         PROP. VIS.&I.R. WAVES IN ATMOS., P5-6. DIMENSIONS OF VAR-
C         IABLES FOLLOWS.
C
C         SIGMA IN KM-1     E IN MB              T IN K               GNU IN CM-1
C         P IN MB           TAU IN C             EO IN MB             C IN K
C         B IS NO DIM.      CO IN MOL-1 M2 ATM-1                      CON1 IN MOL G-1
C         CON2 IN M KM-1    SIGMA IN KM-1        A IN G M-3 ATM
C
          DIMENSION P(40),T(40),DPDEP(40),ALT(40),ABSRP(20),SIGMA(40)
          DIMENSION CO2(20,5),NFNAM(12)
          TYPE " DISC FILE NAME ?"
          READ(11,3000) NFNAM(1)
3000      FORMAT(S12)
          OPEN 1,NFNAM,ATT="I",ERR=100
          ACCEPT "NUMBER OF LINES IN FILE ?", N1
          DO 4 I=1,N1
          READ FREE(1) P(I),ALT(I),T(I),DPDEP(I)
          ALT(I)=ALT(I)/1000.
          T(I)=T(I)+273.16
4         CONTINUE
          CALL CLOSE(1,IER)
          TYPE "WHICH ATMMOSPHERIC ABSORPTION MODEL TO USE ?"
          TYPE "1=TROP 2=MS, 3=MW, 4=SS, 5=SW  "
          ACCEPT MODE
          CON1=3.327E22
          CON2=1.E3
          EO=6.11
          B=7.5
          C=237.3
          GAMMA=0.002
          GNU=944.194
          DO 10 I=1,N1
          IF(DPDEP(I).LT.100.)GO TO 8
          SIGMA(I)=0.0
          GO TO 10
8         TAU=T(I)-DPDEP(I)-273.16
          E=EO*10.**(B*TAU/(C+TAU))
          CO=(1.25E-26+2.34E-23*EXP(-.0083*GNU))*EXP(1800.*(1./T(I)-1./296.))
          A=217.*E*(E+GAMMA*(P(I)-E))/(T(I)*1013.)
          SIGMA(I)=CON1*CON2*CO*A
10        CONTINUE
C
C         INTEGRATE ABSORPTION, INTERPOLATING BETWEEN RAWINSONDE
C         LEVELS AND ACCOUNT FOR PARTIALS (SIG AND SIGLO) AT
C         BEGINNING AND END OF EVEN KM INTERVALS.
```

```
C       BE SURE RAWINSONDE DATA GOES AT LEAST ONE LEVEL ABOVE 10 KM.
C
        N=1
        NL=1
        ABSRP(1)=0.0
        SIGLO=0.0
        ABTOT=0.0
        WRITE(12,3001) NFNAM(1)
3001    FORMAT(" OUTPUT FOR  RAWINSONDE FILE ",S12,/)
        WRITE (12,1001)
        DO 20 I=2,10
        ABSRP(I)=SIGLO
25      IF(ALT(N).GT.I) GO TO 27
        N=N+1
        GO TO 25
27      NH=N-1
        DO 28 K=NL,NH
        ABSRP(I)=ABSRP(I)+(SIGMA(K)+SIGMA(K+1))*(ALT(K+1)-ALT(K))/2.
28      CONTINUE
        SIG=SIGMA(N)-(I-ALT(N))*(SIGMA(N)-SIGMA(N+1))/(ALT(N+1)-ALT(N))
        SIGLO=(SIGMA(N+1)+SIG)*(ALT(N+1)-I)/2.
        ABSRP(I)=ABSRP(I)+(I-ALT(N))*(SIGMA(N)+SIG)/2.
        ABTOT=ABTOT+ABSRP(I)
        DBLOSS=8.6859*ABTOT
        NL=N+1
1001    FORMAT("  ALT.     ALPHA   2-WAY LOSS",/,"  (KM)     (KM-1)
     + (DB)",/)
        WRITE(12,1000) I,ABSRP(I),DBLOSS
1000    FORMAT(I5,F10.5,F7.2)
20      CONTINUE
        STOP
100     TYPE "ERROR IN OPENING RAWINSONDE FILE"
        STOP
        END
```

APPENDIX B

LISTING OF PROGRAM BETA

```
      COMMON/DD/ARAY(2048)
      COMMON/SIGNL/ PWR(1000), IDAT(2048), AVGPWR(2048)
      DIMENSION PARM(25), IHEAD(200), XABSRP(30,3)
      DIMENSION BEAM(30),ABSRP(30,3),PWRAV(1000),AVEP(100)
      EQUIVALENCE (PARM(1),IHEAD(51)),(AVEP(1),IDAT(1))
      LOGICAL ISW
      DATA BEAM/1.081, 1.366, .921, .688, .551, .450,
     + .405, .356, .331, .313, .300, .295, .280, .275,
     + .265, .260, .256, .25, .246, .241, .24, .239, .239, .238,
     + .237, .237, .236, .236, .235, .235/
      DATA XABSRP/  .3669, .2310, .1386, .0819, .0592, .0475, ; TROP
     + .0382, .0305, .0248, .0200, .0156, .0123, .0096, .0072,
     + .0054, .0040, .0032, .0034, .0041, .0049, .0059, .0070,
     + .0079, .0086, .0093, .0121, .0121, .0121, .0121, .0121,
     + .2459, .1519, .0979, .0685, .0526, .0424,        ;MIDLAT SUM
     + .0351, .0287, .0232, .0187, .0148, .0116, .0090, .0080,
     + .0080, .0080, .0080, .0080, .0081, .0084, .0087, .0091,
     + .0096, .0102, .0106, .0118, .0118, .0118, .0118, .0118,
     + .0689, .0568, .0473, .0385, .0309, .0251,        ;MIDLAT WIN
     + .0204, .0164, .0132, .0117, .0104, .0090, .0088, .0086,
     + .0085, .0083, .0081, .0101, .0102, .0077, .0077, .0077,
     + .0077, .0077, .0081, .0081, .0081, .0081, .0081, .0081/
C
C             THIS PROGRAM LOOKS AT RAW SIGNALS AND COMPUTES BACK-
C             SCATTERING FOR DATA COLLECTED WITH THE NOAA COHERENT
C             DOPPLER LIDAR.  MAXIMUMS:  1000 RECORDS/FILE, 2048
C             POINTS/RECORD, 30 KM RANGE.  PROGRAMMED BY M. J.
C             POST AND RON RICHTER.
C
C             ARRAY BEAM CONTAINS BEAM APODIZATION FACTORS FROM TRL
C             (F = 99 KM, E-2 RADIUS = 0.15 M, MATCHED TX AND LO.
C             GOOD TO 11 KM.  INCREMENTS ARE 1 KM IN LIDAR RANGE
C             (NOT ASL).  CHANGED VALUES FOR 2-30 KM RANGE 3/15/83.
C
C             ARRAY ABSRP CONTAINS P20 ABSORPTION
C             COEFFICIENTS FROM MJP, 8/5/82.
C             INCREMENTS ARE 1 KM ASL.
C             BECOMES THE 2-WAY PATH-INTEGRATED
C             ABSORPTION ARRAY LATER IN THE PROGRAM.
C
      CALL VMEM(II,IER)
      IF(II.LT.12) TYPE "NOT ENOUGH MEMORY, SEE D. DAVIS"
      IF(II.LT.12) STOP
      CALL MAPDF(12,ARAY,4,2,IER)
      CALL INITT(30)
      CALL ANMODE
      ICURF=0
      ICURR=0
      ICHAN=22
      CAL=1.0
      ACCEPT "TAPE UNIT NUMBER ",ITU
      CALL MTINIT(ITU,ICHAN,INTER,IOPER)
C
C             SNR EQUATION CONSTANTS AND FACTORS FOLLOW.
C
1      CALL CALIB(CAL)
      TYPE
      ACCEPT "BANDWIDTH (IN MHZ) ",BW
      BW=BW*1.E6
      HNU=1.88E-20
      D=0.3
      RLAM=1.059E-5
C
```

```
C           ETA CHANGED FROM 0.024 TO 0.048 ON 2/3/83 BY MJP TO
C           ACCOUNT FOR SULPHUR DEPOLARIZATION.
C
      ETA=0.048
      PI=3.14159
      C=2.998E8
      P1=8.*HNU*BW
      P2=((PI*D**2)/(4.0*RLAM))**2
      IPLOT = 0
      TYPE
      ACCEPT "ALTITUDE OF LIDAR (KM ASL) ",ALTLID
      TYPE
      ACCEPT "INITIAL RANGE (KM) ",RANGE1
      ACCEPT "END RANGE (KM) ",RANGE2
      TYPE
      TYPE "WHICH ATMOSPHERIC ABSORPTION MODEL TO USE ?"
      TYPE "1=TROPICAL"
      TYPE "2=MIDLATITUDE SUMMER"
      ACCEPT "3=MIDLATITUDE WINTER  ", MOD
C
C           POSITION TAPE AT CORRECT FILE AND RECORD
C
      TYPE
      ACCEPT "FILE NUMBER ? ", IFN
      ACCEPT "START RECORD NUMBER ? ", IREC
      ACCEPT "NUMBER OF RECORDS TO PROCESS ? ", NREC
      TYPE
      ACCEPT"ZENITH ANGLE OF SCANNER (DEGREES) ? ",THETA
      TYPE
      ACCEPT "PER PULSE ENERGY (J) WAS ? ",EJ
      TYPE
      ACCEPT "CONTINUE (1) OR GO BACK AND START OVER (0)? ",IANS
      CALL NEWPAG
      IF(IANS.EQ.0) GO TO 1
      CALL TAPOS(ICHAN,IFN,IREC,ICURF,ICURR,IHEAD,IDAT)
      IF(IHEAD(7).NE.IFN) TYPE"FILE # READ DOES NOT MATCH FILE REQUESTED"
C
C           INITIALIZE THE ABSORPTION COEFFICIENT ARRAY FOR
C           THE GIVEN LIDAR ALTITUDE (ALTLID) ABOVE SEA LEVEL
C           (ASL) AND SCANNER ZENITH ANGLE (THETA).
C
      PRF=1.E6/PARM(5)
      P3=PI*ETA*EJ*C*D**2
      ZZZ=THETA*0.017453293
      ZZZ=COS(ZZZ)
      IALT=ALTLID
      IALT1=IALT+1
      DIFALT=IALT1-ALTLID
      IF(IALT.LE.0) GO TO 15
      DO 10 I=1,IALT
            ABSRP(I,MOD)=0.0
10    CONTINUE
15    IF(IALT1.LE.0) GO TO 20
      ABSRP(IALT1,MOD)=XABSRP(IALT1,MOD)*DIFALT
20    IALT2 = IALT1+1
      DO 25 KL=IALT2,30
            ABSRP(KL,MOD) = XABSRP(KL,MOD)
25    CONTINUE
      CUMK=0.0
C
C           INTEGRATE THE ABSORPTION COEFFICIENTS FOR ROUND-TRIP
C           LOSS TO ALL RANGES.
C
```

```
      DO 30 I=1,30
               CUMK=CUMK+ABSRP(I,MOD)/ZZZ
               ABSRP(I,MOD)=EXP(-2.*CUMK)
30    CONTINUE
      BFLAG=0.
C
C              INITIALIZE POWER ARRAYS
C
      DO 35 I=1,2048
               AVGPWR(I) = 0.0
35    CONTINUE
      DO 40 I=1,1000
               PWR(I) = 0.0
               PWRAV(I)=0.0
40    CONTINUE
C
C              SET SCALING OF PLOTS.
C
45    TYPE"SET SWITCH (0): UP FOR AVERAGE PLOTS ONLY."
      TYPE"                DOWN FOR SINGLE RECORD PLOTS."
      TYPE
      TYPE "NOTE:  SWITCH (0) MAY BE CHANGED DURING PROCESSING."
      TYPE
      TYPE
      TYPE "THE FOLLOWING SCALING OPTIONS CANNOT BE CHANGED."
      TYPE
50    ACCEPT" LINEAR (1) OR DB (0) SCALING ? ", LINDB
      IF(LINDB.EQ.0.OR.LINDB.EQ.1) GO TO 55
      GO TO 50
55    TYPE
      ACCEPT" HI (1) OR LO (0) SCALING ? ", LOHI
      IF(LOHI.EQ.0.OR.LOHI.EQ.1) GO TO 60
      GO TO 55
60    TYPE
      TYPE
      TYPE"PLOTTING FORMAT IS SET FOR:"
      TYPE
      IF(ISW(0)) TYPE"    AVERAGE PLOTS ONLY"
      IF(.NOT.ISW(0)) TYPE"     SINGLE RECORDS, THEN AVERAGE PLOTS"
      IF(LOHI.EQ.1.AND.LINDB.EQ.1) TYPE"     X-AXIS 0-20000 LINEAR"
      IF(LOHI.EQ.1.AND.LINDB.EQ.0) TYPE"     X-AXIS 0-40 DB"
      IF(LOHI.EQ.0.AND.LINDB.EQ.1) TYPE"     X-AXIS 0-400 LINEAR"
      IF(LOHI.EQ.0.AND.LINDB.EQ.0) TYPE"     X-AXIS 0-12 DB"
      TYPE
      TYPE
      TYPE
      ACCEPT "IF THIS FORMAT IS OK, ENTER 1.  IF NOT, ENTER 0.  ", LSW
      IF(LSW.EQ.0) CALL NEWPAG
      IF(LSW.EQ.0) GO TO 45
C
C              PRINT TAPE HEADER
C
      CALL NEWPAG
      IF(PARM(1).GT.10.0.OR.PARM(1).LT.0.08) ACCEPT "DIGITIZER RATE ?"
     + , PARM(1)
      WRITE(10,1010) IFN,(IHEAD(I),I=1,6)
      WRITE(10,1011) PARM(1)
      WRITE(10,1012) PARM(2)
      WRITE(10,1013) PARM(3)
      WRITE(10,1014) PARM(4)
      WRITE(10,1015) PARM(5)/1000.
      WRITE(10,1016) PARM(10)
      PAUSE "STRIKE ANY KEY TO CONTINUE"
```

```
1010    FORMAT(/," FILE ",I2,
     +  /," DATE ",I2,"/",I2,"/",I2,/" TIME ",I2,":",I2,":",
     +  I2,//)
1011    FORMAT(1X," 1",5X,"DIGITIZER RATE (.08-10MHZ) ",T45,F8.2)
1012    FORMAT(1X," 2",5X,"DELAY TO 1ST SAMPLE (US) "T45,F8.2)
1013    FORMAT(1X," 3",5X,"NUMBER OF SAMPLES PER PULSE ",T45,F8.2)
1014    FORMAT(1X," 4",5X,"TRIGGER MODE ",T45,F8.2)
1015    FORMAT(1X," 5",5X,"TRIGGER SPACING (MS) ",T45,F8.2)
1016    FORMAT(1X,"10",5X,"NUMBER OF RECORDS DIGITIZED. ",T45,F8.2)
C
C             CALCULATE START,STOP INDICES
C             FROM TAPE HEADER INFORMATION
C
        DELAY = PARM(2)
        DELT = 1.0/PARM(1)
        DELR = .149896*DELT
        DELZ = DELR*ZZZ
        RR1=(DELAY-2.0)*.149896
        STTIM = RANGE1 /.149896 - DELAY + 2.0
        ISTART = INT( STTIM/DELT + 0.5)
        SPTIM = RANGE2/.149896 - DELAY + 2.0
        ISTOP = INT( SPTIM/DELT + 0.5)
        IF(ISTOP.GT.2048) ISTOP = 2048
        IF(ISTOP-ISTART+1 .GT. 2048) ISTART = ISTOP-2048+1
        IF(ISTART.LT.1) ISTART = 1
        NPTS = ISTOP - ISTART + 1
        IF(ISTART.EQ.1) RANGE1=RR1
        IF(ISTOP.EQ.2048) RANGE2 = RR1 + NPTS/PARM(1) * 0.149896
        R1 = RANGE1*ZZZ + ALTLID
        R2 = RANGE2*ZZZ + ALTLID
C
C             LOOP THRU REQUIRED NUMBER OF RECORDS (NREC) TO
C             READ IN ALL REQUIRED DATA.
C
        ICMD = 2048
        DO 110 K=1,NREC
              KREC=K+IREC-1
              CALL MTDIO(ICHAN,ICMD,IDAT,ISTAT,IWRD,IER)
              IF(IER.NE.1) GO TO 215
C
C             DECODE R AND Q FROM IDAT ARRAY (USING R2Q2 FUNCTION )
C             CALCULATE POWERS AND SUM.  REMOVE SPIKES.
C
              KM1=K-1
              TKM1=KM1/10.
              JJ = 1
              IF(K.GT.4) GO TO 75
              DO 70 I=ISTART,ISTOP
                    RQ=R2Q2(I)
                    PWR(K)=PWR(K)+RQ
                    ARAY(JJ)=RQ
                    AVGPWR(JJ)=AVGPWR(JJ)+RQ
                    JJ=JJ+1
70            CONTINUE
              GO TO 85
75            DO 80 I=ISTART,ISTOP
                    RQ=R2Q2(I)
                    THRESH=AVGPWR(JJ)/TKM1
                    IF(RQ.GE.THRESH)RQ=AVGPWR(JJ)/KM1
                    PWR(K)=PWR(K)+RQ
                    ARAY(JJ)=RQ
                    AVGPWR(JJ)=AVGPWR(JJ)+RQ
              JJ = JJ + 1
```

```
80            CONTINUE
C
C             SET  SCALING AND PLOT SINGLE RECORDS IF REQUESTED
C
85            IF(ISW(0)) GO TO 105
              IPLOT = 1
              IF(LINDB.EQ.1) GO TO 95
              DO 90 I=1,NPTS
                   PPWR=ARAY(I)
C                  CALL VFETCH(PPWR,I)
C                  IF(PPWR.LT.1.0) PPWR=1.0
                   IF(PPWR.EQ.1.)ARAY(I)=1.
C                  IF(PPWR.EQ.1.0) CALL VSTASH(PPWR,I)
                   PPWR = 10.0 * ALOG10(PPWR)
                   ARAY(I)=PPWR
90            CONTINUE
              IF(LOHI.EQ.1) YMAX = 40.0
              IF(LOHI.EQ.0) YMAX = 12.0
              GO TO 100
95            IF(LOHI.EQ.1) YMAX = 20000.0
              IF(LOHI.EQ.0) YMAX = 400.0
              CALL GRAFR(ARAY,NPTS,IFN,IHEAD,R1,R2,YMAX,
     +        IPLOT,KREC,1,DELZ,IREC,0)
              GO TO 105
100           CALL GRAFR(ARAY,NPTS,IFN,IHEAD,R1,R2,YMAX,
     +        IPLOT,KREC,1,DELZ,IREC,0)
105           PWR(K) =PWR(K)/NPTS
110      CONTINUE
         IF(LINDB.EQ.1) GO TO 120
         DO 115 K=1,NREC
              PWR(K)=10.0*ALOG10(PWR(K))
115      CONTINUE
120      KREC=KREC-IREC+1
C
C             CALCULATE AVERAGES AND PLOT
C
         DO 125 I=1,NPTS
              AVGPWR(I)=AVGPWR(I)/NREC
125      CONTINUE
         IF(LINDB.EQ.1) GO TO 135
         DO 130 I=1,NPTS
              AVPWDB = 10.0*ALOG10(AVGPWR(I))
              ARAY(I)=AVPWDB
130      CONTINUE
135      IF(LINDB.EQ.0.AND.LOHI.EQ.1) YMAX = 40.0
         IF(LINDB.EQ.0.AND.LOHI.EQ.0) YMAX = 12.0
         IF(LINDB.EQ.1.AND.LOHI.EQ.1) YMAX = 20000.0
         IF(LINDB.EQ.1.AND.LOHI.EQ.0) YMAX = 400.0
         XMAX = NREC/PRF
C
C             PLOT TIME HISTORY OF INTEGRATED POWER PER RECORD.
C
         CALL GRAFT(PWR,NREC,IFN,IHEAD,XMAX,YMAX,IREC,R1,R2)
         ICURR=IREC+NREC
         ACCEPT "CONTINUE (1) OR REPOSITION TAPE TO RE-RUN (0)? ",IANS
         CALL NEWPAG
         IF(IANS.EQ.0)GO TO 1
         IPLOT=0
         IF(LINDB.EQ.0)GO TO 140
C
C             PLOT RAW POWER DATA AVERAGED OVER ALL PROCESSED RECORDS.
C
         CALL GRAFR(AVGPWR,NPTS,IFN,IHEAD,R1,R2,YMAX,
```

```
     +   IPLOT,KREC,1,DELZ,IREC,0)
         GO TO 145
140      CONTINUE
         CALL GRAFR(ARAY(1),NPTS,IFN,IHEAD,R1,R2,YMAX,
     +   IPLOT,KREC,1,DELZ,IREC,0)
C
C                  OPTION TO PRINT DATA
C
145      ACCEPT "PRINT RAW DATA ?   YES = 1  NO = 0  ", IANS
         IREND=IREC+NREC-1
         IF(IANS.NE.1) GO TO 150
         PRINT 1017, IFN,IREC,IREND,(IHEAD(I),I=1,6)
1017     FORMAT(1H1,"FILE ",I2,/," RECS ",I3," THRU ",I4,
     +   /," DATE ",I2,"/",I2,"/",I2,
     +   /," TIME ",I2,":",I2,":",I2,//)
         PRINT 1018
1018     FORMAT(" AVERAGE POWER VS. RANGE",//)
         PRINT 1019,(I,AVGPWR(I),I=1,NPTS)
1019     FORMAT(5(8X,I3,1X,F10.2))
C
C                  CHOOSE A NOISE WINDOW TO FIND MEAN NOISE LEVEL (AVNOIS)
C
150      ACCEPT "BEG, END ALTITUDES (KM) FOR NOISE WINDOW ?  ",AN1,AN2
         RNOS1=(AN1-ALTLID)/ZZZ
         RNOS2=(AN2-ALTLID)/ZZZ
         TNOS1=RNOS1/.149896-DELAY+2.0
         NOS1=INT(TNOS1/DELT+0.5)
         TNOS2=RNOS2/.149896-DELAY+2.0
         NOS2=INT(TNOS2/DELT+0.5)
         IF(NOS1.LT.1)NOS1=1
         IF(NOS2.GT.2048)NOS2=2048
         NNOIS=NOS2-NOS1+1
         IF (NNOIS .GT. 2048) NOS1 = NOS2 - 2048 + 1
         IF (NNOIS .GT. 2048) NNOIS = 2048
         AVNOIS=0.
         NOS1 = NOS1 - ISTART
         NOS2 = NOS2 - ISTART
         IF( NOS1 .LT. 1 ) NOS1 = 1
         DO 155 I=NOS1,NOS2
                  AVNOIS=AVNOIS+AVGPWR(I)
155      CONTINUE
         AVNOIS=AVNOIS/NNOIS
C
C                  BEGIN CALCULATING BETA  SNR'S FOR GIVEN ALTITUDE
C                  AVERAGING INTERVALS.
C
         TYPE
         ACCEPT "ALTITUDE RESOLUTION IN KM (1.005 FOR ARCHIVING) ?", DALT
         TYPE
         NINC=DALT/DELZ
         ZINC=NINC*DELZ
         NN=67*PARM(1)/10.
         QUAL=1./SQRT(NN)+1./SQRT(NINC)
         WRITE(10,1020) QUAL
1020     FORMAT("Q-FACTOR MUST BE GREATER THAN",F5.2)
         ICOUNT=0
         DO 165 K=1,NPTS,NINC
                  ICOUNT=ICOUNT+1
                  AVSIG=0.
                  DO 160 I=1,NINC
                           AVSIG=AVSIG+AVGPWR(K-1+I)
160               CONTINUE
                  AVSIG=AVSIG/NINC
```

```
              AVEP(ICOUNT)=AVSIG
              ALT=R1+(ICOUNT-1)*ZINC+ZINC/2.
              SIGNOS=AVNOIS/SQRT(NREC)
              QFAC=(AVSIG-AVNOIS)/SIGNOS
              CALL VSTASH(QFAC,4096+ICOUNT)
              SNR=(AVSIG-AVNOIS)/AVNOIS
              IF(SNR.LT.0.0) SNR = 0.0
              PWR(ICOUNT)=SNR
              RNG = (ALT-ALTLID)/ZZZ
              R=RNG*1000.
C
C             ADD LIDAR ALTITUDE ASL FOR
C             INDEXING THE ABSORPTION ARRAY ONLY.
C             LINEARLY INTERPOLATE THE BEAM ARRAY
C             IN RANGE AND THE ABSORPTION ARRAY IN ALTITUDE
C
              IRNG=RNG
              DIFRAN=RNG-IRNG
              IRNG1=IRNG+1
              IRA=ALT
              IRA1=IRA+1
              DRANG=ALT-IRA
              RKAP=(BEAM(IRNG1)-BEAM(IRNG))*DIFRAN+BEAM(IRNG)
        ABSP=(ABSRP(IRA1,MOD)-ABSRP(IRA,MOD))*DRANG+ABSRP(IRA,MOD)
              IF(IRNG.GE.30) RKAP=BEAM(30)
              IF(IRA.GE.30) ABSP=ABSRP(30,MOD)
              P4=P1*(R**2+P2)/(P3*RKAP*ABSP)
C
C             USE AVPWDB ARRAY AS THE BETA PROFILE ARRAY
C
              BETA=P4*SNR*CAL
              IF(QFAC.LT.QUAL) BETA=1.E-15
              CALL VSTASH(BETA,2048+ICOUNT)
165     CONTINUE
        ICM1=ICOUNT-1
C
C             ELIMINATE ISOLATED "VALID" BETA VALUES
C
        DO 170 II=2,ICM1
              CALL VFETCH(BEF,II+2047)
              CALL VFETCH(AFT,II+2049)
        IF(BEF.EQ.1.E-15.AND.AFT.EQ.1.E-15) CALL VSTASH(BEF,II+2048)
170     CONTINUE
        TYPE
C
C             PRINTOUT OPTION
C
        ACCEPT "PRINT OUT BETA PROFILE ?   YES = 1    NO = 0   ", IANS
        IF(IANS.NE.1) GO TO 180
        PRINT 1017,IFN,IREC,IREND,(IHEAD(I),I=1,6)
        PRINT 1021
1021    FORMAT ("  ALT (KM)   BETA(M-1 SR-1)   Q-FACTOR      SNR",/)
        DO 175 J=1,ICOUNT
              ALT=(J-1)*ZINC+ZINC/2.+R1
              CALL VFETCH(AVPWDB,J+2048)
              CALL VFETCH(PWRDB,J+4096)
              PRINT 1022, ALT,AVPWDB,PWRDB,PWR(J)
1022          FORMAT(F7.1,7X,1PE9.2,6X,0PF6.2,F12.3)
175     CONTINUE
C
C             PLOT BETA PROFILE
C
180     DO 185 I=1,ICOUNT
```

```
      CALL VFETCH(AVPWDB,I+2048)
      IF(AVPWDB.LE.0.) AVPWDB=1.E-20
      AVPWDB=10.*ALOG10(AVPWDB)+110.
      CALL VSTASH(AVPWDB,I+2048)
185   CONTINUE
      YMAX=40.
      BFLAG=1.
      DO 190 III=1,ICOUNT
      CALL VFETCH(ARAY(III),2048+III)
190   CONTINUE
      CALL GRAFR(ARAY,ICOUNT,IFN,IHEAD,R1,R2,YMAX,
     + IPLOT,KREC,1,ZINC,IREC,BFLAG,AN1,AN2,MOD)
      TYPE
      ACCEPT "ANOTHER NOISE WINDOW (1) OR CONTINUE (0) ? ",IANS
      IF(IANS.EQ.1) GO TO 150
C
C           ARCHIVE BETA DATA IF DESIRED, CHECKING FOR PROPER
C           POSITIONING AND INCREMENTING.
C
      TYPE
      ACCEPT "ARCHIVE THE DATA ?  YES=1  NO=0 ",IANS
      IF(IANS.NE.1) GO TO 210
      IALT=ALT
      DIFALT=ALT-IALT
      IF(DALT.EQ.1.005.AND.DIFALT.LT..2) GO TO 195
      TYPE
      TYPE"ERROR IN ALTITUDE INCREMENT OR INDEXING"
      TYPE"DALT SHOULD BE 1.005 , INDEXING ON EVEN KM (ASL)."
      TYPE
      GO TO 150
195   DO 200 I=1,ICOUNT
      CALL VFETCH(AVPWDB,2048+I)
      AVPWDB=10.**((AVPWDB-110.)/10.)
      CALL VSTASH(AVPWDB,2048+I)
200   CONTINUE
      CALL APPEND(3,"ARCH",0,IER)
      IF(IER.NE.1) STOP - ARCH OPEN ERROR
      WRITE BINARY(3) IFN,IREC,IREND,(IHEAD(I),I=1,6),ICOUNT,
     + MOD,R1,DALT,ALTLID
      DO 205 I=1,ICOUNT
      ALT=(I-1)*ZINC+ZINC/2.+R1
      IALT=ALT
      IALT1=IALT+1
      DIFALT=ALT-IALT
      IF(DIFALT.GE.0.5)IALT=IALT1
      ALT=IALT
      CALL VFETCH(AVPWDB,2048+I)
      CALL VFETCH(PWRDB,4096+I)
      WRITE BINARY(3) ALT,AVPWDB,PWRDB,PWR(I),AVEP(I)
205   CONTINUE
      CLOSE 3
      CALL NEWPAG
      TYPE"BE SURE TO RECORD THIS ARCHIVING ON THE MASTER LIST"
      WRITE(10,1025)(IHEAD(I),I=1,6),IFN,NREC,BW/1.E6,EJ
1025  FORMAT(/,"DATE",T25,I2,"/",I2,"/",I2,//"TIME",T25,I2,":",
     + I2,":",I2,//"FILE NUMBER",T25,I8,//"NUMBER OF RECORDS",T25,
     + I8,//"BANDWIDTH (MHZ)",T25,F8.1,//"LASER  POWER",T25,F8.3///)
210   ICURR=IREC+NREC
      ACCEPT "PROCESS ANOTHER FILE (1) OR STOP (0)? ",IANS
      IF(IANS.EQ.1) CALL NEWPAG
      IF(IANS.EQ.1) GO TO 1
      CALL MTRWN (ICHAN)
      STOP
```

```
C                 TAPE ERROR ROUTINE
C
215      CALL NEWPAG
         WRITE(10,1023) IER,ISTAT,IWRD
         WRITE(10,1024) KREC
1023     FORMAT(1X,"TAPE ERROR",/,"IER=",I4,"ISTAT= ",B16," IWRD ",I5)
1024     FORMAT(1X,"TAPE ERROR OCCURED ON RECORD NUMBER ",I5)
220      CALL MTRWN (ICHAN)
         STOP
         END
```

REFERENCES

Lawrence, T. R., 1983. Comparison of the observed antenna gain properties of the NOAA lidar with theory. To be published in Applied Optics.

Lott, G. A., 1976. Precipitable water over the United States, Vol. 1, Monthly means. NOAA Tech. Rep. NWS-20, 174 pp.

McClatchey, R. A., R. W. Fenn, J. E. A. Selby, F. E. Volz, and J. S. Garing, 1972. Optical properties of the atmosphere (Revised). AFCRL-72-0497, 108 pp.

Post, M. J., R. A. Richter, R. M. Hardesty, T. R. Lawrence, and F. F. Hall, Jr., 1982. NOAA's pulsed, coherent, IR Doppler lidar--characteristics and data. Proc. Soc. Photo-Opt. Instrum. Eng. 300:60.

Roberts, R. E., J. E. A. Selby, and L. M. Biberman, 1976. Infrared continuum absorption by atmospheric water vapor in the 8-12 μm window. Appl. Opt. 15:2085-2089.

Zuev, V. Ye., 1972. Propagation of Visible and Infrared Waves in the Atmosphere. NASA Technical Translation NASA TT F-707, pp. 5-6.